

An algorithm to restore spectral signatures of all inherent optical properties of seawater using a value of one property at one wavelength

Vladimir I. Haltrin

Naval Research Laboratory, Ocean Sciences Branch, Code 7331
Stennis Space Center, MS 39529-5004, USA, e-mail: <haltrin@nrlssc.navy.mil>

ABSTRACT

A one-parameter chlorophyll-based optical model of inherent optical properties of seawater is used to develop an algorithm to express the whole set of spectral inherent optical properties through the one optical parameter – the beam attenuation coefficient at one wavelength. This model is valid for the beam attenuation coefficient between 0 and 10 m^{-1} in visible part of spectrum (400-720 nm). The FORTRAN code to compute optical properties of seawater through a beam attenuation coefficient at one wavelength is released for public use.

Keywords: Ocean optics, inherent optical properties, scattering, absorption, seawater.

1.0 INTRODUCTION

This work is based on a one-parameter optical model of inherent sea optical properties presented previously at the Ocean Optics XIV meeting in Hawaii (Haltrin, 1998a). The one-parameter optical model expresses the whole set of spectral inherent optical properties, the absorption and scattering coefficients and the phase function of scattering, through one parameter – the concentration of chlorophyll. The model is valid for the values of chlorophyll concentration ranging from 0 to 12 mg / m^3 . The model was used to obtain values of inherent optical properties in the range of wavelengths between 400 and 720 nm over the range of chlorophyll concentrations cited. The resulting data sets were statistically processed to obtain wavelength-dependent regressions that couple different inherent optical properties with the beam attenuation coefficient.

2.0. MODEL OF SEAWATER OPTICAL PROPERTIES

All optical properties of seawater are divided into two groups: inherent optical properties and apparent optical properties. Inherent optical properties depend only on processes of absorption and single scattering in seawater. Apparent optical properties depend on inherent optical properties, geometry of illumination, and processes of transmission and reflection by the sea surface and sea bottom.

There are only two major inherent optical properties: the absorption coefficient a and the angular scattering coefficient $\beta(\vartheta)$ (which is a function of a scattering angle ϑ). All other inherent optical properties, the phase function of scattering $p(\vartheta) \equiv \beta(\vartheta) / b$, the scattering coefficient $b = 0.5 \int_0^\pi \beta(\vartheta) \sin \vartheta d\vartheta$, the beam attenuation coefficient $c = a + b$, the single scattering albedo $\omega_0 = b / c$, the backscattering coefficient $b_B = 0.5 \int_{\pi/2}^\pi \beta(\vartheta) \sin \vartheta d\vartheta$, the probability of backscattering $B = 0.5 \int_{\pi/2}^\pi p(\vartheta) \sin \vartheta d\vartheta$, are derivatives of these two major inherent optical properties (Haltrin and

Kattawar, 1993). The model presented here is based on timer-proven parts proposed by different researchers (See Refs. in Haltrin, 1998a).

The absorption coefficient $a(\lambda)$, (m^{-1}) is taken to be:

$$a(\lambda) = a_w(\lambda) + a_c^0(\lambda) \left(C_c / C_c^0 \right)^{0.602} + a_f^0 C_f \exp(-k_f \lambda) + a_h^0 C_h \exp(-k_h \lambda), \quad (1)$$

where $a_w(\lambda)$ is the pure water absorption coefficient in m^{-1} , λ is the wavelength of light in nm, $a_c^0(\lambda)$ is the specific absorption coefficient of chlorophyll in m^2 / mg , C_c is the total concentration of chlorophyll in mg / m^3 ($C_c^0 = 1 \text{ mg} / \text{m}^3$) (Prieur and Satyendranath, 1981), $a_f^0 = 35.959 \text{ m}^2 / \text{mg}$ is the specific absorption coefficient of fulvic acid; $k_f = 0.0189 \text{ nm}^{-1}$; $a_h^0 = 18.828 \text{ m}^2 / \text{mg}$ is the specific absorption coefficient of humic acid; $k_h = 0.01105 \text{ nm}^{-1}$; C_f and C_h are, respectively, concentrations of fulvic and humic acids in mg / m^3 (Carder *et al.*, 1989).

The scattering $b(\lambda)$ and backscattering $b_B(\lambda)$ coefficients are calculated according to:

$$b(\lambda) = b_w(\lambda) + b_s^0(\lambda) C_s + b_l^0(\lambda) C_l, \quad b_B(\lambda) = 0.5 b_w(\lambda) + B_s b_s^0(\lambda) C_s + B_l b_l^0(\lambda) C_l, \quad (2)$$

here

$$B_s = 0.5 \int_{\pi/2}^{\pi} p_s(\vartheta) \sin \vartheta d\vartheta = 0.039, \quad B_l = 0.5 \int_{\pi/2}^{\pi} p_l(\vartheta) \sin \vartheta d\vartheta = 0.00064, \quad (3)$$

B_s is a probability of backscattering by small particles, B_l is a probability of backscattering by large particles, $b_w(\lambda)$ is the scattering coefficient by pure water in m^{-1} , $b_s^0(\lambda)$ and $b_l^0(\lambda)$ are, respectively, the specific scattering coefficients in m^2 / g for small and large particulate matter, C_s and C_l are, respectively, concentrations in g / m^3 of small and large particles. The equation for $b_w(\lambda)$ is derived by interpolating the data published by Morel and Prieur (1977):

$$b_w(\lambda) = (5.826 \cdot 10^{-3} \text{ m}^{-1}) (400/\lambda)^{4.322}. \quad (4)$$

The spectral dependencies for scattering coefficients of small and large particulate matter are given by the following equations (Kopelevich, 1983; Haltrin 1985):

$$b_s^0(\lambda) = (1.1513 \text{ m}^2 / \text{g}) (400/\lambda)^{1.7}, \quad b_l^0(\lambda) = (0.3411 \text{ m}^2 / \text{g}) (400/\lambda)^{0.3}. \quad (5)$$

The phase function of scattering is derived by Kopelevich (1983) from results of *in situ* measurements. The Kopelevich model expresses the total hydrosol scattering function as a linear combination of two phase functions p_s and p_l . The phase function p_s describes scattering by small terrigenous fraction of particles with the density $\rho_s = 2 \text{ g} / \text{cm}^3$. The phase function p_l describes scattering by large particles associated with a biogenic fraction of marine hydrosol with the density $\rho_l = 1 \text{ g} / \text{cm}^3$. The total hydrosol angular scattering coefficient is expressed as follows:

$$\beta_H(\lambda, \vartheta) = b_s^0(\lambda) p_s(\vartheta) C_s + b_l^0(\lambda) p_l(\vartheta) C_l. \quad (6)$$

The phase functions in Eq. (6) can be expressed by the following regressions (Haltrin, 1997):

$$p_s(\vartheta) = 5.61746 \exp\left(\sum_{n=1}^5 s_n \vartheta^{3n/4}\right), \quad p_l(\vartheta) = 188.381 \exp\left(\sum_{n=1}^5 l_n \vartheta^{3n/4}\right), \quad (7)$$

here ϑ is the scattering angle in degrees. The coefficients s_n and l_n are given in Haltrin (1997, 1998a) and are built into functions `fp_small` and `fp_large` of APPENDIX A.

The seawater angular scattering coefficient is a linear combination of a Rayleigh phase function of scattering p_R and a hydrosol phase functions p_s and p_l :

$$\beta(\lambda, \vartheta) = b_w(\lambda) p_R(\vartheta) + b_s^0(\lambda) p_s(\vartheta) C_s + b_l^0(\lambda) p_l(\vartheta) C_l, \quad p_R(\vartheta) = 0.7823 + 0.6531 \cos^2 \vartheta \quad (8)$$

Equations (1)–(11) allow us to compute inherent optical properties of seawater as functions of wavelength and five concentrations C_c , C_h , C_f , C_s , C_l of dissolved and suspended matter.

Results of *in situ* measurements of seawater optical properties show that in a majority of cases any two formally independent optical properties correlate with each other (case I waters). Morel (1980) and Gordon and Morel (1983) propose the following correlation to estimate the chlorophyll concentration C_r in the upper ocean layer:

$$C_r = 1.92 I_c^{1.8}, \quad I_c = R(550)/R(440), \quad (9)$$

here $R(\lambda)$ is a diffuse reflectance of the sea at wavelength λ . All these dependencies indicate that we can choose a single parameter to characterize all inherent optical properties. The optical model given by Eqs. (1)-(8) depends on five parameters: C_c , C_h , C_f , C_s , C_l . The one-parameter model implies that we can express any four of these concentrations through a chosen fifth one. It is convenient to choose the chlorophyll concentration C_c as our main parameter. To derive four dependencies that express four concentrations C_c , C_f , C_s , C_l through a chlorophyll concentration C_c we minimized the following five-dimensional functional:

$$\Delta(C_c, C_f, C_h, C_s, C_l) = |C_c - C_r| \left| C_c - 1.92 [R_\infty(550)/R_\infty(440)]^{1.8} \right|. \quad (10)$$

The dependence of R_∞ on a and b_B in Eq. (9) was taken from Haltrin (1998b). Two additional stabilizing relationships were used to restrict a number of solutions. For chlorophyll concentrations in the range of $0 \leq C_c \leq 12 \text{ mg/m}^3$ several solutions were found. The single physically meaningful solution to this problem is given in a form of the four dependencies (Haltrin 1998a):

$$\left. \begin{aligned} C_f &= 1.74098 \cdot C_c \cdot \exp(0.12327 \cdot C_c), \quad C_s = 0.01739 \cdot C_c \cdot \exp(0.11631 \cdot C_c), \\ C_h &= 0.19334 \cdot C_c \cdot \exp(0.12343 \cdot C_c), \quad C_l = 0.76284 \cdot C_c \cdot \exp(0.03092 \cdot C_c). \end{aligned} \right\} \quad (11)$$

These dependencies allow calculation of the concentrations of dissolved organic matter (C_f , C_h) and concentrations of suspended particles (C_s , C_l) through the concentration of chlorophyll C_c . Eqs. (11) and Eqs. (1)-(8) constitute a one-parameter model of seawater optical properties. The validation of this model based on independent experimental data is given in Haltrin (1998a).

3.0. ALGORITHM TO CALCULATE INHERENT OPTICAL PROPERTIES THROUGH A VALUE OF BEAM ATTENUATION COEFFICIENT AT ONE WAVELENGTH

Equations (11) give a one-parameter model of seawater optical properties. Existence of a solution to Eq. (11) means that any chosen inherent optical property may be used as a parameter. For example, if we choose the beam attenuation, we can formulate a one-parameter model of sea optical properties based on a $c(\lambda)$. For example, let us use $\Delta c(\lambda)$, the difference between beam attenuation coefficients of seawater and pure water,

$$\Delta c(\lambda) = c(\lambda) - a_w(\lambda) - b_w(\lambda), \quad 0 \leq \Delta c \leq 10 \text{ m}^{-1}, \quad (12)$$

as a regression parameter, here $a_w(\lambda)$ is taken from Pope and Fry (1997) and $b_w(\lambda)$ is given by Eq. (5). In this case it is possible to derive the following regression relationships between $\Delta c(\lambda)$ and wavelength λ and chlorophyll concentration C_c and other optical properties a , b , and b_B :

$$C_c = \Delta c(c_1 + \Delta c(c_2 + c_3 \Delta c)), \quad c_j = \sum_{n=0}^5 c_{jn} \left(\frac{\lambda - 400}{350} \right)^n, \quad j = 1, 2, 3, \quad (13)$$

$$a = a_w + \Delta c(a_1 + \Delta c(a_2 + a_3 \Delta c)), \quad a_j = \sum_{n=0}^5 a_{jn} \left(\frac{\lambda - 400}{350} \right)^n, \quad j = 1, 2, 3, \quad (14)$$

$$b = b_w + \Delta c(b_1 + \Delta c(b_2 + b_3 \Delta c)), \quad b_j = \sum_{n=0}^5 b_{jn} \left(\frac{\lambda - 400}{350} \right)^n, \quad j = 1, 2, 3, \quad (15)$$

$$b_b = 0.5 b_w + \Delta c(g_1 + \Delta c(g_2 + g_3 \Delta c)), \quad g_j = \sum_{n=0}^5 g_{jn} \left(\frac{\lambda - 400}{350} \right)^n, \quad j = 1, 2, 3. \quad (16)$$

Equations (13)-(16) are derived and valid for the wavelengths in the interval $410 \leq \lambda \leq 715$ nm. The coefficients $c_j, a_j, b_j, g_j, j=1,2,3$ are given in file "coef.in" of APPENDIX B. Equations (13)-(16) allow us to derive a chlorophyll concentration C_c , absorption, scattering, and backscattering coefficients a , b , and b_b for the same wavelength λ that corresponds to the value of c used to derive parameter Δc . To restore the whole set of spectral and angular optical properties we should use the approach given in Section 2.

4.0 ACKNOWLEDGMENT

The author thanks continuing support at the Naval Research Laboratory through the programs SS 5939-A8 and LOE 6640-08. This article represents NRL contribution PP/7331-98-0038.

APPENDIX A: A PROGRAM TO COMPUTE OPTICAL PROPERTIES OF SEAWATER THROUGH A BEAM ATTENUATION COEFFICIENT AT ONE WAVELENGTH

```

!      program      IOPregr (written by Vladimir I. Haltrin <haltrin@nrlssc.navy.mil>
!      this program is free for any non-commercial use, a reference to this article is required
implicit none
integer      i,j,Nang,Nan
parameter   (Nang=361)
real*8      c,lam,cw,bw,dc,x,Cc,Cf,Ch,Cs,C1,a,b,bB,wl
real*8      ai,bi,bBi,bpw,btot,thet,A0,bs,bl,gi,ci
real*8      c1s,c2s,c3s, a1s,a2s,a3s,b1s,b2s,b3s, g1s,g2s,g3s
real*8      c1(0:5),c2(0:5),c3(0:5), a1(0:5),a2(0:5),a3(0:5)
real*8      b1(0:5),b2(0:5),b3(0:5), g1(0:5),g2(0:5),g3(0:5)
real*8      lm(35),aw(35),Ac0(35),at(35),bt(35),bBt(35)
real*8      p(35,Nang),ang(Nang),fawater,fbwater,fphfunct
logical     warning,fbadinput
character   tb

open(11, file = 'IOPregr.in', status='old')
  read(11,*) c
  read(11,*) lam
close(11)

open(12, file = 'coef.in', status='old')
  read(12,*) (c1(i), i=0,5)
  read(12,*) (c2(i), i=0,5)
  read(12,*) (c3(i), i=0,5)
  read(12,*) (a1(i), i=0,5)
  read(12,*) (a2(i), i=0,5)
  read(12,*) (a3(i), i=0,5)
  read(12,*) (b1(i), i=0,5)
  read(12,*) (b2(i), i=0,5)
  read(12,*) (b3(i), i=0,5)
  read(12,*) (g1(i), i=0,5)
  read(12,*) (g2(i), i=0,5)
  read(12,*) (g3(i), i=0,5)
close(12)

open(13, file='lawac35.in', status='old')
  read(13,'(//)')
  do i = 1, 35
    read(13,*) lm(i),aw(i),Ac0(i)
  end do
close(13)

open(14, file='ang.in', status='old')
  read(14,*) Nan ! Nan < Nang
  read(14,*) (ang(j), j=1,Nan)
close(14)

cw = fawater(aw,lam) + fbwater(lam)
warning = fbadinput(cw,lam,c)

```

```
dc = c-cw
x = (lam-400.)/350.
bw = fbwater(lam)
call scoeff(c1,c2,c3,x, c1s,c2s,c3s)
Cc = dc*(c1s+dc*(c2s+c3s*dc)) ! chlorophyll concentration
call scoeff(a1,a2,a3,x, a1s,a2s,a3s)
a = fawater(aw,lam)+dc*(a1s+dc*(a2s+a3s*dc)) ! absorption coefficient
call scoeff(b1,b2,b3,x, b1s,b2s,b3s)
b = bw + dc*(b1s+dc*(b2s+b3s*dc)) ! scattering coefficient
call scoeff(g1,g2,g3,x, g1s,g2s,g3s)
bB = 0.5*bw+dc*(g1s+dc*(g2s+g3s*dc)) ! backscattering coefficient
call scalcons(Cc, Cf,Ch,Cs,Cl)
do i = 1,35
  wl = lm(i)
  A0 = Ac0(i)
  call siopsi(Cc,Cf,Ch,Cs,Cl,wl,A0, bs,bl,ai,bi,bBi)
  at(i) = aw(i) + ai
  bpw = fbwater(wl)
  btot = bpw + bi
  bt(i) = btot
  bBt(i) = 0.5*bpw + bBi
  do j=1,Nan
    thet = ang(j)
    p(i,j) = fphfunct(bpw,bs,bl,btot,thet)
  end do
end do
tb = CHAR(9)
open(21, file='ioppr.out', status='new')
  if (warning) then
    write(21,8)
    write(21,9)
  end if
  write(21,30) 'Attenuation coefficient, c = ',c, ' 1/m'
  write(21,30) 'Input wavelength, lam = ',lam, ' nm'
  write(21,30) 'Computed Chlorophyll concentration, Cc = ',Cc
&   ', ' mg/m^3'
  write(21,30) 'Computed absorption coefficient, a = ',a, ' 1/m'
  write(21,30) 'Computed scattering coefficient, b = ',b, ' 1/m'
  write(21,30) 'Computed backscattering coefficient, bB = ',bB
&   ', ' 1/m'
close(21)
open(22, file='iops.out', status='new')
  write(22,40) 'lam',tb,'c',tb,'a',tb,'b',tb,'bB',tb,'g'
  do i = 1, 35
    ai = at(i) ! absorption coefficient
    bi = bt(i) ! scattering coefficient
    ci = ai+bi ! attenuation coefficient
    bBi = bBt(i) ! backscattering coefficient
    gi = bBi/(ai+bBi) ! Gordon's parameter
    write(22,50) lm(i),tb,ci,tb,ai,tb,bi,tb,bBi,tb,gi
  end do
close(22)
open(23, file='phfs.out', status='new',recl=465)
  write(23,60) 'ang\lam', (tb,lm(i),i=1,35)
  do j = 1, Nan
    write(23,70) ang(j), (tb,p(i,j),i=1,35)
  end do
close(23)
8 format(' Bad input parameters in the file "IOPregr.in"')
9 format(' -----')
```

```
30 format (a42, f10.6,a7)
40 format (a5, 5(a1,a12))
50 format (f5.1, 5(a1,g12.5))
60 format (a7, 35(a1,f12.1))
70 format (f7.3, 35(a1,g12.5))
end

subroutine scalcons(Cc, Cf,Ch,Cs,Cl)
implicit none
real*8 Cc, Cf,Ch,Cs,Cl
Cf = 1.74098*Cc*EXP(0.12327*Cc)
Ch = 0.19334*Cc*EXP(0.12343*Cc)
Cs = 0.01739*Cc*EXP(0.11631*Cc)
Cl = 0.76284*Cc*EXP(0.03092*Cc)
return
end

real*8 function fbwater(lam)
implicit none
real*8 lam
fbwater = 5.826E-3*(400./lam)**4.322
return
end

real*8 function fawater(aw,lam)
implicit none
integer i
real*8 aw(35),lam,lm
i = INT(lam-370.)/10
lm = 370.+ 10.*i
fawater = aw(i) + 0.1*(aw(i+1)-aw(i))*(lam-lm)
return
end

logical function fbadinput(cw,lam,ce)
implicit none
real*8 ce,lam,cw
fbadinput = .false.
if (ce.lt.cw) then
ce = cw
fbadinput = .true.
end if
if (lam.lt.410.) then
lam = 410.
fbadinput = .true.
end if
if (lam.gt.715.) then
lam = 715.
fbadinput = .true.
end if
return
end

subroutine scoeff(v1,v2,v3,x, w1,w2,w3)
implicit none
real*8 v1(0:5),v2(0:5),v3(0:5),x, w1,w2,w3
w1 = v1(0)+x*(v1(1)+x*(v1(2)+x*(v1(3)+x*(v1(4)+x*v1(5))))))
w2 = v2(0)+x*(v2(1)+x*(v2(2)+x*(v2(3)+x*(v2(4)+x*v2(5))))))
w3 = v3(0)+x*(v3(1)+x*(v3(2)+x*(v3(3)+x*(v3(4)+x*v3(5))))))
return
end

subroutine siopsi(Cc,Cf,Ch,Cs,Cl,lam,A0, bs,bl,ai,bi,bBi)
implicit none
real*8 Cc,Cf,Ch,Cs,Cl,lam,A0, bs,bl,ai,bi,bBi,r,la
r = 400./lam
```

```

bl = 0.3410739*Cl*(r**0.3) ! scattering coeff. by large fraction
bs = 1.1513020*Cs*(r**1.7) ! scattering coeff. by small fraction
la = 450.-lam
ai = A0*0.06*(Cc**0.602) + Cf*0.00728*EXP(0.01890*la)
& + Ch*0.13040*EXP(0.01105*la) ! absorption coeff. by species
bi = bs + bl ! total scattering by suspensions
bBi = 0.039*bs + 0.00064*bl ! total backscatt. by suspensions
return
end

real*8 function fphfunct(bpw,bs,bl,btot,ang)
implicit none
real*8 bpw,bs,bl,btot,ang,beta
real*8 fpwater,fpsmall,fplarge
beta = bpw*fpwater(ang)+bs*fpsmall(ang)+bl*fplarge(ang)
fphfunct = beta/btot
return
end

real*8 function fpwater(ang) ! ang in degrees
implicit none
real*8 ang,mu
mu = COSD(ang)
fpwater = 0.7823 + 0.6531*mu*mu
return
end

real*8 function fpsmall(ang) ! ang in degrees
implicit none
real*8 ang,x
x = ang**0.75
x = x*(-2.957089e-2+x*(-2.78294e-2+x*(1.255406e-3+x
& *(-2.15588e-5+1.356632e-7*x))))
fpsmall = 52.39389*EXP(x) ! corrected 6/14/00
return
end

real*8 function fplarge(ang) ! ang in degrees
implicit none
real*8 ang,x
x = ang**0.75
x = x*(-1.604327+x*(8.157686e-2+x*(-2.150389e-3+x
& *(2.419323e-5-6.578455e-8*x))))
fplarge = 7653.704*EXP(x) ! corrected 6/14/00
return
end

```

APPENDIX B: INPUT FILES TO THE PROGRAM IOPREGR

File "coef.in":

2.422697E+0	3.857256E+0	-1.578237E+1	6.585905E+1	-1.051054E+2	5.427571E+1
-1.877098E-1	-9.457076E-1	8.166927E+0	-3.267484E+1	4.923368E+1	-2.459349E+1
5.642058E-3	1.237942E-1	-1.166176E+0	4.416329E+0	-6.444667E+0	3.163730E+0
2.885964E-1	-6.317607E-1	3.383521E+0	-1.538335E+1	2.537777E+1	-1.331364E+1
1.697134E-2	1.295791E-2	-1.476434E+0	6.401668E+0	-9.702480E+0	4.841352E+0
2.505272E-5	-1.152725E-2	2.177258E-1	-8.421014E-1	1.219591E+0	-5.943660E-1
7.114036E-1	6.317610E-1	-3.383523E+0	1.538335E+1	-2.537777E+1	1.331364E+1
-1.697134E-2	-1.295790E-2	1.476434E+0	-6.401669E+0	9.702482E+0	-4.841353E+0
-2.505272E-5	1.152725E-2	-2.177258E-1	8.421014E-1	-1.219591E+0	5.943660E-1
2.497181E-3	-1.861928E-3	-5.878085E-5	2.063419E-2	-4.438652E-2	2.562029E-2
2.772841E-4	1.384905E-3	-1.299578E-3	-3.868762E-3	8.154464E-3	-4.274115E-3
-1.109959E-5	-5.498079E-5	-6.231651E-5	9.251807E-4	-1.554635E-3	7.728582E-4

file "IOPreg.in":

```

0.2      <-- beam attenuation coefficient in 1/m (0 ≤ c ≤ 10 1/m)
450.    <-- wavelength in nm (410 ≤ lam ≤ 715)

```


File "ang.in":

```
30 <-- Nan, a number of angles for phase function (in degrees); ang(Nan) (Nan < 362):
0. .1 0.3 0.5 1.0 1.5 3.0 5. 10 15. 20. 25. 30. 35. 40. 45.
50. 60. 70. 80. 90. 100. 110. 120. 130. 140. 150. 160. 170. 180.
```

File "lawac35.in":

Aw according to Pope & Fry (1997)	540.0	0.04740	0.416
Ac0 according to Prieur & Satyendranath	550.0	0.05650	0.357
lam	560.0	0.06190	0.294
Aw	570.0	0.06950	0.276
Ac0	580.0	0.08960	0.291
380.0	590.0	0.13510	0.282
390.0	600.0	0.22240	0.236
400.0	610.0	0.26440	0.252
410.0	620.0	0.27550	0.276
420.0	630.0	0.29160	0.317
430.0	640.0	0.31080	0.334
440.0	650.0	0.34000	0.356
450.0	660.0	0.41000	0.441
460.0	670.0	0.43900	0.595
470.0	680.0	0.46500	0.502
480.0	690.0	0.51600	0.329
490.0	700.0	0.62400	0.215
500.0	710.0	0.82700	0.160
510.0	720.0	1.23100	0.112
520.0			
530.0			

REFERENCES

- K. L. Carder, R. G. Stewart, G. R. Harvey, and P. B. Ortner, "Marine humic and fulvic acids: Their effects on remote sensing of ocean chlorophyll," *Limnol. Oceanogr.* **34**, pp. 68-81 (1989).
- H. R. Gordon, and A. Morel, *Remote assessment of ocean color for interpretation of satellite visible imagery* (A Review, Springer Verlag, 1983).
- V. I. Haltrin (a.k.a. В. И. Халтурин), "Propagation of Light in the Sea Depth," in *Optical Remote Sensing of the Sea and the Influence of the Atmosphere*, V. A. Urdenko and G. Zimmermann, eds., pp. 20-62, in Russian, (German Democratic Republic Academy of Sciences Institute for Space Research, Berlin, 1985).
- V. I. Haltrin, "One-parameter model of seawater optical properties," in *Ocean Optics XIV CD-ROM*, Kailua-Kona, Hawaii, 10-13 November 1998, published by Office of Naval Research, (1998a).
- V. I. Haltrin, "Self-consistent approach to the solution of the light transfer problem for irradiances in marine waters with arbitrary turbidity, depth and surface illumination," *Appl. Optics*, **37**, pp. 3773-3784 (1998b).
- V. I. Haltrin, "Apparent optical properties of the sea illuminated by Sun and sky: case of optically deep sea," *Appl. Optics*, **37**, 8336-8340, (1998c).
- V. I. Haltrin, "Theoretical and empirical phase functions for Monte Carlo calculations of light scattering in seawater," in *Proceedings of the Fourth International Conference Remote Sensing for Marine and Coastal Environments*, **I**, pp. 509-518, (Envir. Res. Inst. of Michigan, Ann Arbor, MI, 1997).
- V. I. Haltrin, and G. W. Kattawar, "Self-consistent solutions to the equation of transfer with elastic and inelastic scattering in oceanic optics: I. Model," *Appl. Optics*, **32**, pp. 5356-5367, (1993).
- O. V. Kopelevich, "Small-Parametric Model of the Optical Properties of Seawater," in *Ocean Optics, I: Physical Ocean Optics*, A. S. Monin, ed., pp. 208-234, in Russian., Nauka Publishing, Moscow, (1983).
- A. Morel, "In-water and remote measurement of ocean color," *Bound. Layer Meteor.*, **18**, pp. 177-201 (1980).
- A. Morel, and L. Prieur, "Analysis of Variations in Ocean Color," *Limnol. Oceanogr.*, **22**, pp. 709-722 (1977).
- Pope R. M., and E. S. Fry, "Absorption spectrum (380-700 nm) of pure water: II. Integrating cavity measurements," *Appl. Optics*, **36**, pp. 8710-8723 (1997).
- L. Prieur, and S. Satyendranath, "An Optical Classification of Coastal and oceanic waters based on the specific spectral absorption curves of phytoplankton pigments, dissolved organic matter, and other particulate materials," *Limnol. Oceanogr.*, **26**, pp. 671-689 (1981).
- V. A. Timofeyeva, "Optical characteristics of turbid media of the seawater type," *Izv. Atmos. Ocean Physics*, **7**, pp. 863-865, (1971).